Laboratory Report # 6

Name: *Cyril Andre B. Durango* **Date Completed:** *November 8, 2024*

Laboratory Exercise Title: *Behavioral Modeling of Sequential Circuits*

Target Course Outcomes:

CO1: Create descriptions of digital hardware components such as in combinational and sequential circuits using synthesizable Verilog HDL constructs.

CO2: Verify the functionality of HDL-based components through design verification tools.

Exercise 6A | JK Flip-flop. Create a behavioral description of a JK flip-flop using either if-else statements or a case statement.

Requirements:

1. Clk: negative-edged triggered
2. Reset: active high asynchronous reset



Figure 1. Block Diagram of a JK Flipflop

Table 1. JK Flipflop Characteristic Table

Reset	Clk	J	K	Q
1	x	x	x	0
0	↑	x	x	Q
0	↓	0	0	Q
0	↓	0	1	0
0	↓	1	0	1
0	↓	1	1	Q'

```
/**************************************************************
 * AUTHOR:      CYRIL ANDRE B. DURANGO
 * CLASS:       CPE 3101L - INTRODUCTION TO HDL
 * SCHEDULE:    Group 4 | FRIDAY | 7:30 AM - 10:30 AM
 * --------------------------------------------------------
 * PROJECT FILE TITLE: JK_Flipflop.v
 *
 **************************************************************/
module JK_Flipflop
(
    input wire  J, K, Reset, Clk,
    output reg  Q,
    output      Q_bar
);
    assign Q_bar = ~Q;
    always @(negedge Clk, posedge Reset)
    begin

        if (Reset)
            Q <= 1'b0;
        else
            begin
                if (!J & !K)
                    Q <= Q;

                else if (!J & K)
                    Q <= 1'b0;

                else if (J & !K)
                    Q <= 1'b1;

                else
                    Q <= ~Q;

            end
    end

endmodule
```

Figure 2. JK_Flipflop.v Verilog File



Figure 3. Compilation report for JK_Flipflop.v

```verilog
/****************************************************************
 * AUTHOR:      CYRIL ANDRE B. DURANGO
 * CLASS:       CPE 3101L - INTRODUCTION TO HDL
 * SCHEDULE:    Group 4 | FRIDAY | 7:30 AM - 10:30 AM
 * ------------------------------------------------------
 * PROJECT FILE TITLE: tb_JK_Flipflop.v
 *
****************************************************************/
`timescale 1 ns / 1 ps
module tb_JK_Flipflop();
   reg   J, K, Reset, Clk;
   wire  Q, Q_bar;

   JK_Flipflop UUT(J, K, Reset, Clk, Q, Q_bar);

   initial
      Clk = 1'b0;
   always
      #4 Clk = ~ Clk;

   initial begin
      Reset = 1'b1; #35
      Reset = 1'b0;
   end

   initial begin
      J = 1'b0;   K = 1'b0;   #10
      J = 1'b0;   K = 1'b1;   #10
      J = 1'b1;   K = 1'b0;   #8
      J = 1'b1;   K = 1'b1;   #8

      J = 1'b0;   K = 1'b0;   #8
      J = 1'b0;   K = 1'b1;   #8
      J = 1'b1;   K = 1'b0;   #8
      J = 1'b1;   K = 1'b1;   #8
   $stop;
   end

endmodule
```

Figure 4. tb_JK_Flipflop.v Verilog File



Figure 5. JK Flipflop Generated Waveform

The following generated waveform simulates the outputs of a JK Flipflop given the simulated inputs. The active high reset is enabled first to set the outputs of Q be 0 and Q_bar be 1. Despite switching the J and K signals, the Q and Q_bar outputs stay 0 and 1 respectively due to the active-high reset.

Once the reset becomes 0, and once the system sees a negative edge on the clock (blue signal), it outputs a corresponding Q based on the JK inputs. It is much evident on the last 2 clock cycles. In the first cycle, J is 1, K is 0, therefore Q is 1. In the next cycle, J is 1, K is 1, therefore Q is the negation of the previous value. Since previous value was 1, the value displayed is 0.

Exercise 6B | 4-bit Binary Up/Down Counter. Design a 4-bit binary up/down counter using behavioral description.

Requirements:

1. Clk: negative-edged triggered
2. nReset: active low asynchronous reset

Table 2. 4-bit Binary Up/Down Counter Truth Table

nReset	Clk	Load	Count_in	Count_en	Up	Count_out
0	x	x	x	x	x	0
1	↑	1	C	x	x	C
1	↓	0	x	0	x	Hold
1	↓	0	x	1	0	Count down
1	↓	0	x	1	1	Count up
1	↓	0	x			

```
/**************************************************************
 * AUTHOR:     CYRIL ANDRE B. DURANGO
 * CLASS:      CPE 3101L - INTRODUCTION TO HDL
 * SCHEDULE:   Group 4 | FRIDAY | 7:30 AM - 10:30 AM
 * ----------------------------------------------------------
 * PROJECT FILE TITLE: BinaryCounter_4Bit.v
 *
**************************************************************/

module BinaryCounter_4Bit
(
   input wire           Count_en, nReset, Clk, Load, Up,
   input        [3:0] Count_in,
   output reg   [3:0] Count_out

);

   always @(negedge Clk, negedge nReset)
   begin
      if (!nReset)
         Count_out <= 4'd0;

      else if (Load)
         Count_out <= Count_in;

      else if (Count_en && Up)
         Count_out <= Count_out + 1;

      else if (Count_en && !Up)
         Count_out <= Count_out - 1;

   end

endmodule
```

Figure 6. BinaryCounter_4bit.v Verilog File



Figure 7. Compilation Report for BinaryCounter_4bit.v

```verilog
/*************************************************************
 * AUTHOR:     CYRIL ANDRE B. DURANGO
 * CLASS:      CPE 3101L - INTRODUCTION TO HDL
 * SCHEDULE:   Group 4 | FRIDAY | 7:30 AM - 10:30 AM
 *
 * --------------------------------------------------------
 * PROJECT FILE TITLE: tb_BinaryCounter_4Bit.v
 *
*************************************************************/

`timescale 1 ns/1 ps
module tb_BinaryCounter_4Bit();
   reg         Count_en, nReset, Clk, Load, Up;
   reg   [3:0] Count_in;
   wire  [3:0] Count_out;

   BinaryCounter_4Bit UUT(Count_en, nReset, Clk, Load, Up, Count_in, Count_out);

   initial
      Clk = 1'b0;

   always
      #2 Clk = ~ Clk;

   initial begin
      nReset = 1'b0; #67
      nReset = 1'b1;
   end

   initial begin
      Load = 1'b1;    #35
      Load = 1'b0;    #32
      Load = 1'b1;    #32
      Load = 1'b0;
   end

   initial begin
      Count_en = 1'b0;   #19
      Count_en = 1'b1;   #16
      Count_en = 1'b0;   #16
      Count_en = 1'b1;   #16
      Count_en = 1'b0;   #16
      Count_en = 1'b1;   #16
      Count_en = 1'b0;   #16
      Count_en = 1'b1;
   end

initial begin
   Count_in = 4'd8;
   Up = 1'b0;   #9
   Up = 1'b1;   #10
   Up = 1'b0;   #8
   Up = 1'b1;   #8
   Up = 1'b0;   #8
   Up = 1'b1;   #8
   Up = 1'b0;   #8
   Up = 1'b1;   #8

   Up = 1'b0;   #8
   Up = 1'b1;   #8
   Up = 1'b0;   #8
   Up = 1'b1;   #8
   Up = 1'b0;   #8
   Up = 1'b1;   #8
   Up = 1'b0;   #16
   Up = 1'b1;   #16


$stop;
end


endmodule
```

Figure 8. tb_BinaryCounter_4bit.v Verilog File



Figure 9. Generated waveform of BinaryCounter_4bit.v (First Half)

In Figure 9, it can be seen here that the output of the 4-bit bidirectional counter stays at 0 since the nReset is active-low triggered. Any input signals from Load, Count-en, Up and Count_in are overridden.



Figure 10. Generated waveform of BinaryCounter_4bit.v (Second Half)

In Figure 10, it can be seen here that the output of the 4-bit bidirectional counter stays at 8 in the more than the first half of the figure since the Load (3rd blue signal) is high, and after that is Count_en (4th blue signal) is low. In the first case, it stays at 8 since LOAD is High. In the second case, LOAD is LOW, but Count_en is LOW. Once

Count_en is enabled and that the Up signal (5th blue signal) is low, it starts counting down. Once the Up input signal is High, that is when it starts to count up.

FPGA Verilog Files and Outputs

```
/****************************************************************
 * AUTHOR:     CYRIL ANDRE B. DURANGO
 * CLASS:      CPE 3101L - INTRODUCTION TO HDL
 * SCHEDULE:   Group 4 | FRIDAY | 7:30 AM - 10:30 AM
 * ------------------------------------------------------
 * PROJECT FILE TITLE: Clock_converter.v
 *
****************************************************************/

module Clock_converter(
   input wire  Clk_FPGA,
   output reg  Clk_converted
);

   reg   [25:0]   counter;

   always @(posedge Clk_FPGA)
   begin
      counter <= counter + 1;
      if (counter == 25'd25000000) begin
         Clk_converted <= ~Clk_converted;
         counter <= 0;
      end
   end

endmodule
```

Figure 11. Clock_converter.v Verilog File

```
/****************************************************************
 * AUTHOR:     CYRIL ANDRE B. DURANGO
 * CLASS:      CPE 3101L - INTRODUCTION TO HDL
 * SCHEDULE:   Group 4 | FRIDAY | 7:30 AM - 10:30 AM
 * ------------------------------------------------------
 * PROJECT FILE TITLE: JK_Flipflop_FPGA.v
 * PROJECT DESCRIPTION:
 *
****************************************************************/

module JK_Flipflop_FPGA(
   input wire  J, K, Reset, Clk_FPGA,
   output wire Q, Q_bar
);

   wire Clk;

   Clock_converter Clk1(Clk_FPGA, Clk);
   JK_Flipflop JKFF1(J, K, Reset, Clk, Q, Q_bar);


endmodule
```

Figure 12. JK_Flipflop_FPGA.v Verilog File (Exercise 6A)



Figure 13. I/O Components Used in Exercise 6A

Table 3. I/O Assignments for Activity 6A

Verilog Name	FPGA Pin Assignment	Corresponding I/O Device	Description
J	PIN_C10	SW0	J and K input signals
K	PIN_C11	SW1	
Reset	PIN_D12	SW2	Resets the flipflop
Q_bar	PIN_A8	LEDR0	Inverse Flipflop Output
Q	PIN_A9	LEDR1	Flipflop Output
CLK_FPGA	PIN_P11	50 MHZ Clock Input	FPGA Clock



Figure 14. Exercise 6A Sample Output

In Figure 14, it can be seen that reset switch and K switch is LOW, and J switch is HIGH . The output Q (LEDR1) is HIGH and its inverse (LEDR0) is LOW

```
/***************************************************************
 * AUTHOR:     CYRIL ANDRE B. DURANGO
 * CLASS:      CPE 3101L - INTRODUCTION TO HDL
 * SCHEDULE:   Group 4 | FRIDAY | 7:30 AM - 10:30 AM
 * -------------------------------------------------------
 * PROJECT FILE TITLE: Binary_4bitcounter_FPGA.v
 *
***************************************************************/

module Binary_4bitcounter_FPGA(
   input wire      Count_en, nReset, Clk_FPGA, Load, Up,
   input wire      [3:0] Count_in,
   output wire     [3:0] Count_out
);

   wire Clk;

   Clock_converter      Clk1(Clk_FPGA, Clk);
   BinaryCounter_4Bit   BinaryCount(Count_en, nReset, Clk, Load, Up, Count_in, Count_out);


endmodule
```

Figure 15. Binary_4bitcounter_FPGA.v Verilog File (Exercise 6B)



Figure 16. I/O Components Used in Exercise 6B

Table 4. I/O Assignments for Activity 6A

Verilog Name	FPGA Pin Assignment	Corresponding I/O Device	Description
Up	PIN_A13	SW6	0 - Count down, 1 = Count Up
Count_en	PIN_A14	SW7	Enable Count
Load	PIN_B14	SW8	Load Count_in
nReset	PIN_F15	SW9	Reset Signal
Count_in[0]	PIN_C10	SW0	Load Inputs
Count_in[1]	PIN_C11	SW1	
Count_in[2]	PIN_D12	SW2	
Count_in[3]	PIN_C12	SW3	
Count_out[0]	PIN_A8	LEDR0	Binary Count Output
Count_out[1]	PIN_A9	LEDR1	
Count_out[2]	PIN_A10	LEDR2	
Count_out[3]	PIN_B10	LEDR3	



Figure 17. I/O Components Used in Exercise 6B

As seen in Figure 17, the input signals (arranged in nReset, Load, Count_en, and Up), nReset and Load are HIGH and the rest of the input signals are LOW. In this configuration, whatever binary count is enabled in the count_in switches will be displayed in the Count_out LEDs